WORKING PAPERS ISSN 0781-4410

SUOMEN AKTUAARIYHDISTYS
The Actuarial Society of Finland

8

Matti Ruohonen

FITTING THE CONVOLUTION OF NEGATIVE BINOMINAL AND POISSON DISTRIBUTIONS ON DATA (1983)

FITTING THE CONVOLUTION OF NEGATIVE BINOMIAL AND POISSON DISTRIBUTIONS ON DATA

by
Matti Ruohonen
Institute for Applied Mathematics
University of Turku, Turku, Finland

ABSTRACT

The purpose of this report is to act as an appendix to the paper [4] of the author. The report contains some proofs and formulae as well as computer programs not included in [4]. The convolution of negative binomial and Poisson distributions was proposed by the author for the distribution of the number of claims in an insurance portfolio.

INTRODUCTION

In this report we present formulae and programs that are needed in computing maximum likelihood estimators for the parameters of the convolution of negative binomial and Poisson distributions. This is a model proposed by the author for the distribution of the number of claims in an insurance portfolio.

Let now N denote the number of claims in one time unit. We take N to be $N = N_1 + N_2$, where N_1 has the Poisson distribution with parameter γ and N_2 has the negative binomial distribution with parameters α and $\beta/(1+\beta)$. The basic formulae needed in this report are

$$P(N=n) = \sum_{k=0}^{n} \frac{\Gamma(k+\alpha)}{\Gamma(\alpha)k!} \frac{\beta^{\alpha}}{(1+\beta)^{k+\alpha}} \frac{\gamma^{n-k} e^{-\gamma}}{(n-k)!} \qquad (1)$$

$$E(N) = \gamma + \alpha/\beta$$
$$Var(N) = \alpha/\beta^2 + \alpha/\beta + \gamma \qquad (2)$$
$$E((N-E(N))^3) = 2\alpha/\beta^3 + 3\alpha/\beta^2 + \alpha/\beta + \gamma$$

and

$$p_o = P(N=0) = (\beta/(1+\beta))^{-\alpha} e^{-\gamma} . \qquad (3)$$

MAXIMUM LIKELIHOOD ESTIMATION FOR THE NEGATIVE BINOMIAL DISTRIBUTION

Because we want to test the hypothesis $H_o:\gamma = 0$ against the alternative $H_1:\gamma > 0$, we first consider the maximum likelihood estimation in the case $\gamma = 0$. Then N has the negative binomial distribution,

$$P(N=n) = \frac{\Gamma(n+\alpha)}{\Gamma(\alpha)n!} \frac{\beta^{\alpha}}{(1+\beta)^{n+\alpha}} . \qquad (4)$$

If we have the observations $n_o, n_1, \ldots, n_k$, where k is the largest number of claims observed, and n_i is the number of risks having had i claims, then the likelihood function can be written as

$$L(\alpha,\beta) = C \left(\frac{\beta}{1+\beta}\right)^{\alpha n} \left(\frac{1}{1+\beta}\right)^{n\bar{x}} \alpha^{n_1+\ldots+n_k} \ldots (\alpha+k-1)^{n_k},$$

where C is a constant. Taking logarithm and derivating with respect to α and β and equating these derivatives to zero we get

$$\alpha = \beta\bar{x} \qquad (5)$$

$$n \ln(\beta/(1+\beta)) + \sum_{j=1}^{k} \left(\sum_{i=j}^{k} n_i\right)(\beta\bar{x}+j-1)^{-1} = 0 . \qquad (6)$$

Because (6) cannot be solved in a closed form we solve it by using Newton's method. For this we need the derivative

of the left hand side of (6). This is

$$n\,(1/\beta - 1/(1+\beta)) + \sum_{j=1}^{k}\Big(\sum_{i=j}^{k} n_i\Big)(-\bar{x})(\beta\bar{x}+j-1)^{-2}.$$

In the following we shall present the FORTRAN-subroutines that are needed in solving (5) and (6). The subroutine PARAM computes the estimator $(\bar{\alpha},\bar{\beta})$ when an initial guess for β is given.

```
      SUBROUTINE PARAM(B,ALFA,BETA,OMEAN,Y,K,SD)
      IMPLICIT DOUBLE PRECISION (A-H,O-Z)
      DIMENSION Y(50)
      EPS=.000000000001
5     CALL FNM(FUN,Y,K,OMEAN,B,SD)
      IF(DABS(FUN)-EPS) 10,10,20
20    CALL DERM(DER,Y,K,OMEAN,B,SD)
      B1=B-FUN/DER
      B=B1
      GO TO 5
10    BETA=B
      ALFA=BETA*OMEAN
      RETURN
      END
```

Here B is the initial guess for β, OMEAN is $\bar{x}$, Y is the vector $(n_o,\ldots,n_k)$, K=k+1 is the number of classes and SD is equal to $n= n_o+\ldots+n_k$.

This program calls the subroutine FNM, which computes the value of the left hand side of (6)

```
      SUBROUTINE FNM(FUN,Y,K,OMEAN,BETA,SD)
      IMPLICIT DOUBLE PRECISION (A-H,O-Z)
      DIMENSION Y(50)
      AB=SD*DLOG(BETA/(1.+BETA))
      SUM1=0.
      DO 20 J=2,K
      SUM2=0.
      DO 30 I=J,K
30    SUM2=SUM2+Y(I)
20    SUM1=SUM1+SUM2/(BETA*OMEAN+J-2.)
      FUN=AB+SUM1
      RETURN
      END
```

It also calls the subroutine DERM, which computes the derivative of the left hand side of (6).

```
      SUBROUTINE DERM(DER,Y,K,OMEAN,BETA,SD)
      IMPLICIT DOUBLE PRECISION (A-H,O-Z)
      DIMENSION Y(50)
      A=SD/BETA-SD/(1.+BETA)
      SUM1=0.
      DO 20 J=2,K
      SUM2=0.
      DO 30 I=J,K
30    SUM2=SUM2+Y(I)
20    SUM1=SUM1-SUM2*OMEAN/(BETA*OMEAN+J-2)**2
      DER=A+SUM1
      RETURN
      END
```

Finally when $(\bar{\alpha},\bar{\beta})$ is obtained we compute the probabilities (4) using the subroutine PROBB.

```
      SUBROUTINE PROBB(P,ALFA,BETA,K)
      IMPLICIT DOUBLE PRECISION (A-H,O-Z)
      DIMENSION P(50)
      P(1)=(BETA/(1.+BETA))**ALFA
      DO 10 I=2,K
10    P(I)=(ALFA+I-2)*P(I-1)/((I-1.)*(1.+BETA))
      RETURN
      END
```

Here the probabilities (4) are collected in the vector P. The statistics needed are computed by the subroutine STATM

```
      SUBROUTINE STATM(K,X,Y,OMEAN,VAR,SKEW,PNOL,SD)
      IMPLICIT DOUBLE PRECISION (A-H,O-Z)
      DIMENSION X(50),Y(50)
      SA=0.
      SB=0.
      SC=0.
      SD=0.
      DO 45 I=1,K
      SA=SA+X(I)*Y(I)
      SB=SB+Y(I)*X(I)**2
      SC=SC+Y(I)*X(I)**3
45    SD=SD+Y(I)
      OMEAN=SA/SD
      VAR=(SB-SD*OMEAN**2)/(SD-1.)
      SKEW=SC/SD-3.*VAR*OMEAN-OMEAN**3
      PNOL=Y(1)/SD
      WRITE(3,60) OMEAN
60    FORMAT (' MEAN ' ,E20.10//)
      WRITE(3,61) VAR
61    FORMAT(' VARIANCE ',E20.10//)
      WRITE(3,62) SKEW
62    FORMAT(' 3. CENTRAL MOMENT', E20.10//)
      WRITE(3,63) PNOL
63    FORMAT(' ZERO CLASS PROBABILITY', E20.10//)
      WRITE(3,64) SD
64    FORMAT (' NUMBER OF OBSERVATIONS', F8.0)
      RETURN
      END
```

The main program is FITNM2.

```
      IMPLICIT DOUBLE PRECISION (A-H,O-Z)
      DIMENSION X(50),Y(50),P(50),TEOR(50)
      CALL ASSIGN (2,'RE:',3)
      CALL ASSIGN (3,'WR:',3)
      WRITE (2,50)
      READ (2,10) K
      DO 35 I=1,K
      WRITE (2,52)
35    READ(2,30) Y(I)
50    FORMAT(' GIVE THE NUMBER OF CLASSES')
52    FORMAT(' GIVE OBSERVATIONS IN THE CLASS')
30    FORMAT (F15.10)
10    FORMAT (I2)
      DO 3 I=1,K
3     X(I)=I-1.
      CALL STATM(K,X,Y,OMEAN,VAR,SKEW,PNOL,SD)
      B=OMEAN/(VAR-OMEAN)
      CALL PARAM(B,ALFA,BETA,OMEAN,Y,K,SD)
      WRITE (3,300) ALFA,BETA
      CALL PROBB(P,ALFA,BETA,K)
      DO 2 I=1,K
2     TEOR(I)=SD*P(I)
300   FORMAT (' ALFA,BETA ', 2E20.10)
      DO 60 I=1,K
60    WRITE (3,65) P(I),TEOR(I),Y(I)
65    FORMAT (E20.10,2F20.2)
      TOSA=SD
      DO 70 I=1,K
      TOSA=TOSA-TEOR(I)
      KL=I
      IF(TOSA-5.) 71,71,70
70    CONTINUE
71    OSA=0.
      DO 72 J=KL,K
72    OSA=OSA+Y(J)
      CHI=0.
      DO 4 I=1,KL-1
4     CHI=CHI+(TEOR(I)-Y(I))**2/TEOR(I)
      TOSA=TOSA+TEOR(KL)
      CHI=CHI+(TOSA-OSA)**2/TOSA
      KL=KL-3
      WRITE (3,5) CHI,KL
5     FORMAT(//' CHI SQUARE IS', F15.4,'  DEGREES OF FREEDOM ARE',I5)
      STOP
      END
```

For other methods of estimation in this case we refer to [1].

MAXIMUM LIKELIHOOD ESTIMATION FOR THE CONVOLUTION

We now turn to our main task: the computing the estimators for the convolution of negative binomial and Poisson distributions. In the likelihood function we write $\eta = \gamma(1+\beta)$ and substitute this for β. The logarithm of the likelihood function is then

$$\tilde{L}(\alpha,\eta,\gamma) = n\alpha \ln \frac{\eta-\gamma}{\eta} - n\gamma + n\bar{x}\ln(\gamma) + \sum_{j=0}^{k} n_j \ln\Big(\sum_{i=0}^{j} \frac{\Gamma(i+\alpha)}{i!(j-i)!\Gamma(\alpha)\eta^i} \Big), \quad (7)$$

see [4]. Denoting

$$w_j(\alpha,\eta) = \sum_{i=0}^{j} \Gamma(i+\alpha)(\Gamma(\alpha)i!(j-i)!\eta^i)^{-1}$$

we can write (7) in the form

$$\tilde{L}(\alpha,\eta,\gamma) = n\alpha \ln \frac{\eta-\gamma}{\eta} - n\gamma + n\bar{x} \ln(\gamma) + \sum_{j=0}^{k} n_j \ln(w_j(\alpha,\eta)), \quad (8)$$

and its partial derivatives as

$$\frac{\partial L}{\partial \gamma} = -n\alpha/(\eta-\gamma) - n + n\bar{x}/\gamma \quad (9a)$$

$$\frac{\partial L}{\partial \alpha} = n \ln((\eta-\gamma)/\eta) + \sum_{j=0}^{k} n_j(\frac{\partial}{\partial\alpha} w_j(\alpha,\eta))/w_j(\alpha,\eta) \quad (9b)$$

$$\frac{\partial L}{\partial \eta} = n\alpha((\eta-\gamma)^{-1} - \eta^{-1}) + \sum_{j=0}^{k} n_j(\frac{\partial}{\partial\eta} w_j(\alpha,\eta))/w_j(\alpha,\eta). \quad (9c)$$

The partial derivatives of w_j here are

$$\frac{\partial}{\partial\alpha} w_j(\alpha,\eta) = ((j-1)!\eta)^{-1} + \sum_{i=2}^{j} (\sum_{l=1}^{i} \prod_{m\neq l}^{i} (\alpha+m-1))/(i!(j-i)!\eta^{i})$$

and

$$\frac{\partial}{\partial\eta} w_j(\alpha,\eta) = - \sum_{i=1}^{j} \Gamma(i+\alpha)/(\Gamma(\alpha)(i-1)!(j-i)!\eta^{i+1}).$$

Because the derivatives (9) are obtained in a closed form we can use a maximization method based on the use of the gradients. The gradient projection method proved to be very slow and inefficient. For this reason the maximization of (8) is performed by Fletcher-Powell-Davidon method, which is a quasi Newton method, see [3].

In the following we shall present the programs used in this procedure.First the subroutine FUNCTM calculates the value of (7) (or equivalently (8)) when the values β and γ are given. The value of α is obtained, when the left hand side

of (9a) is equated to zero, as $\alpha = \beta(\bar{x}-\gamma)$.

```
      SUBROUTINE FUNCTM(Y,BETA,GAMMA,K,OMEAN,FNCT,SD)
      IMPLICIT DOUBLE PRECISION (A-H,O-Z)
      DIMENSION Y(50)
      A=BETA*(OMEAN-GAMMA)
      ALKU=SD*A*DLOG(BETA/(1.+BETA))-SD*GAMMA
      ALKU=ALKU+SD*OMEAN*DLOG(GAMMA)
      D=0.
      DO 20 J=2,K
      B=0.
      C=1.
      DO 30 I=1,J-1
30    C=C/I
      B=B+C
      DO 40 I=1,J-1
      C=(A+I-1)*(J-I)*C/(I*GAMMA*(1.+BETA))
40    B=B+C
      B=Y(J)*DLOG(B)
20    D=D+B
      FNCT=D+ALKU
      RETURN
      END
```

The value of (7) is taken out as FNCT. The symbols used here are the same as in the preceding section. The partial derivatives $\partial\tilde{L}/\partial\alpha$ and $\partial\tilde{L}/\partial\eta$ are calculated by the subroutine GRADAM, when α, η and γ are given. The values are taken out as DFA and DFE.

```
      SUBROUTINE GRADAM(Y,K,OMEAN,ALFA,ETA,GAMMA,DFE,DFA,SD)
      IMPLICIT DOUBLE PRECISION  (A-H,O-Z)
      DIMENSION Y(50),W(50),DWE(50),DWA(50)
      ALKU=SD*ALFA*(1./(ETA-GAMMA)-1./ETA)
      SUM=0.
      DO 20 J=2,K
      BB=0.
      AA=1.
      DO 30 I=1,J-1
30    AA=AA/I
      BB=BB+AA
      DO 70 I=1,J-1
      AA=AA*(ALFA+I-1)*(J-I)/(I*ETA)
70    BB=BB+AA
20    W(J)=BB
      DWE(2)=-ALFA/ETA**2
      DO 40 J=3,K
      BB=0.
      AA=DWE(2)
      DO 50 I=1,J-2
50    AA=AA/I
      BB=BB+AA
      DO 60 I=2,J-1
      AA=AA*(ALFA+I-1.)*(J-I)/(ETA*(I-1.))
60    BB=BB+AA
40    DWE(J)=BB
      DO 80 J=2,K
80    SUM=SUM+Y(J)*DWE(J)/W(J)
      DFE=SUM+ALKU
      DWA(2)=1./ETA
      DO 130 J=3,K
      SUMA=DWA(2)
      DO 105 I=1,J-2
105   SUMA=SUMA/I
```

```
      DO 120 I=2,J-1
      SUML=0.
      DO 110 L=1,I
      TULO=1.
      DO 100 M=1,I
      IF(M-L) 15,100,15
15    TULO=TULO*(ALFA+M-1.)
100   TULO=TULO
110   SUML=SUML+TULO
      CC=1.
      DO 160 M=1,J-1
160   CC=CC/M
      SUML=SUML*CC
      DO 150 M=1,I
150   SUML=SUML*(J-M)/(M*ETA)
120   SUMA=SUMA+SUML
130   DWA(J)=SUMA
      SUMB=0.
      DO 140 I=2,K
140   SUMB=SUMB+Y(I)*DWA(I)/W(I)
      DFA=SUMB+SD*DLOG((ETA-GAMMA)/ETA)
      RETURN
      END
```

The actual maximization is performed by the subroutine MAXB. The optimization is a restricted one. The program MAXB forces the arguments γ and η to stay nonnegative. This means that no penalty function method is used. Additionally no one-dimensional optimization is performed but the step length is halved until a better point is found.

```
      SUBROUTINE MAXB(ALFA,BETA,GAMMA,OMEAN,VAR,PNOL,Y,K,SD)
      IMPLICIT DOUBLE PRECISION (A-H,O-Z)
      DIMENSION Y(50),W(50),DWE(50),DWA(50)
      EPS=.0000001
      H11=1.
      H12=0.
      H21=0.
      H22=1.
      G=OMEAN/2.
      CALL GAB(G,ALFA,BETA,GAMMA,OMEAN,VAR,PNOL)
      BA=BETA
      GA=GAMMA
      IF(GA) 11,11,12
12    IF(BA) 11,11,10
11    GA=OMEAN/2.
      BA=OMEAN/(VAR-OMEAN)
10    ASK=.25*OMEAN
      CALL FUNCTM(Y,BA,GA,K,OMEAN,F1,SD)
      ETA=GA*(BA+1.)
      ALFA=BA*(OMEAN-GA)
      CALL GRADAM(Y,K,OMEAN,ALFA,ETA,GA,DFE,DFA,SD)
13    S1=(H11*DFA+H12*DFE)
      S2=(H21*DFA+H22*DFE)
      DELTA=DSQRT(S1**2+S2**2)
      S2=S2/DELTA
      S1=S1/DELTA
3     A1=ALFA+ASK*S1
      E1=ETA+ASK*S2
      APU=A1+E1-OMEAN
      IF (A1) 1,1,5
5     B1=(APU+DSQRT(APU**2+4.*A1*OMEAN))/(2.*OMEAN)
      G1=A1+E1-B1*OMEAN
```

```
          IF (E1) 1,1,6
 6        IF (B1) 1,1,7
 7        IF (G1) 1,1,8
 8        IF (OMEAN-G1) 1,1,9
 9        CALL FUNCTM(Y,B1,G1,K,OMEAN,F2,SD)
          IF(F2-F1) 1,1,2
 2        CALL GRADAM(Y,K,OMEAN,A1,E1,G1,DFE1,DFA1,SD)
          Q1=-DFA1+DFA
          Q2=-DFE1+DFE
          DENU1=H11*Q1**2+2*Q1*Q2*H21+H22*Q2**2
          OS11=(Q1*H11+Q2*H21)**2
          OS12=(Q1*H11+Q2*H21)*(Q1*H12+Q2*H22)
          OS22=(Q1*H12+Q2*H22)**2
          AN11=-OS11/DENU1
          AN12=-OS12/DENU1
          AN22=-OS22/DENU1
          DENU2=S1*Q1+S2*Q2
          AM11=(ASK*S1**2)/DENU2
          AM12=(ASK*S2*S1)/DENU2
          AM22=(ASK*S2**2)/DENU2
          H11=H11+AM11+AN11
          H12=H12+AM12+AN12
          H21=H12
          H22=H22+AM22+AN22
          BA=B1
          GA=G1
          ETA=GA*(BA+1.)
          ALFA=BA*(OMEAN-GA)
          F1=F2
          DFA=DFA1
          DFE=DFE1
          ASK=0.25*OMEAN
          GO TO 13
 1        ASK=ASK/2.
          IF(ASK-EPS) 4,4,3
 4        BETA=BA
          GAMMA=GA
          ALFA=BA*(OMEAN-GA)
          RETURN
          END
```

The starting point for the maximization is computed by the subroutine GAB, when this gives positive gamma. In the opposite case the starting values are taken as $\gamma = \bar{x}/2$ and $\beta = \bar{x}/(s^2-\bar{x})$. The subroutine GAB solves α, β and γ in fitting $\bar{x}$, s^2 and p_o. The initial guess here is $\gamma = \bar{x}/2$.

```
          SUBROUTINE GAB(G,ALFA,BETA,GAMMA,OMEAN,VAR,PNOL)
          IMPLICIT DOUBLE PRECISION (A-H,O-Z)
          EPS=.0000000001
 5        CALL F(FU,G,PNOL,OMEAN,VAR)
          IF(DABS(FU)-EPS) 10,10,20
 20       CALL DER(D,G,PNOL,OMEAN,VAR)
          IF(DABS(D)-EPS) 10,10,15
 15       G1=G-FU/D
          G=G1
          GO TO 5
 10       GAMMA=G
          BETA=(OMEAN-GAMMA)/(VAR-OMEAN)
          ALFA=(OMEAN-GAMMA)*BETA
          RETURN
          END
```

In the program GAB additional subroutines F and DER are used

```
SUBROUTINE F(FU,GAMMA,PNOL,OMEAN,VAR)
IMPLICIT DOUBLE PRECISION (A-H,O-Z)
A=OMEAN-GAMMA
B=VAR-OMEAN
C=GAMMA+DLOG(PNOL)
D=(A**2)/B
E=DLOG((A+B)/A)
FU=C+D*E
RETURN
END
```

```
SUBROUTINE DER(D,GAMMA,PNOL,OMEAN,VAR)
IMPLICIT DOUBLE PRECISION (A-H,O-Z)
A=OMEAN-GAMMA
B=VAR-OMEAN
C=2.*A/B
D=1.-C*DLOG((A+B)/A)+A/(A+B)
RETURN
END
```

The symbol PNOL here means the zero class probability p_o.
All the statistics needed are given by the subroutine STATM as in the preceding section.
The maximum likelihood estimation and the χ^2-test are performed then by the main program FIPONB

```
      IMPLICIT DOUBLE PRECISION (A-H,O-Z)
      DIMENSION X(50),Y(50),P(50),TEOR(50)
      CALL ASSIGN (2,'RE:',3)
      CALL ASSIGN (3,'WR:',3)
      WRITE (2,50)
      READ (2,10) K
      DO 35 I=1,K
      WRITE (2,52)
35    READ(2,30) Y(I)
50    FORMAT(' GIVE THE NUMBER OF CLASSES')
52    FORMAT(' GIVE OBSERVATIONS IN THE CLASS')
30    FORMAT (F15.10)
10    FORMAT (I2)
      DO 3 I=1,K
3     X(I)=I-1.
      CALL STATM(K,X,Y,OMEAN,VAR,SKEW,PNOL,SD)
      CALL MAXB(ALFA,BETA,GAMMA,OMEAN,VAR,PNOL,Y,K,SD)
      WRITE (3,300) ALFA,BETA,GAMMA
      CALL PROB(P,ALFA,BETA,GAMMA,K)
      DO 2 I=1,K
2     TEOR(I)=SD*P(I)
300   FORMAT (' ALFA,BETA AND GAMMA ', 3E20.10)
      DO 60 I=1,K
60    WRITE (3,65) P(I),TEOR(I),Y(I)
65    FORMAT (E20.10,2F20.2)
      TOSA=SD
      DO 70 I=1,K
      TOSA=TOSA-TEOR(I)
      KL=I
      IF(TOSA-5.) 71,71,70
70    CONTINUE
```

```
71      OSA=0.
        DO 72 J=KL,K
72      OSA=OSA+Y(J)
        CHI=0.
        DO 4 I=1,KL-1
4       CHI=CHI+(TEOR(I)-Y(I))**2/TEOR(I)
        TOSA=TOSA+TEOR(KL)
        CHI=CHI+(TOSA-OSA)**2/TOSA
        KL=KL-4
        WRITE (3,5) CHI,KL
5       FORMAT(//' CHI SQUARE IS', F15.4,'  DEGREES OF FREEDOM ARE',I5)
        STOP
        END
```

The probabilities given by the convolution model, when α, β and γ are known, are given by the subroutine PROB

```
        SUBROUTINE PROB(PR,ALFA,BETA,GAMMA,K)
        IMPLICIT DOUBLE PRECISION (A-H,O-Z)
        DIMENSION PR(50)
        PR(1)=((BETA/(1.+BETA))**ALFA)*DEXP(-GAMMA)
        DO 10 J=2,K
        A=1.
        B=0.
        N=J-1
        DO 15 I=1,N
15      A=A*GAMMA/I
        B=B+A
        DO 16 I=1,N
        A=(ALFA+I-1.)*(N-I+1.)*A/(I*GAMMA*(BETA+1.))
16      B=B+A
10      PR(J)=B*PR(1)
        RETURN
        END
```

Finally the logarithm of the likelihood ratio is computed by the subroutine SUHDEM.

```
        SUBROUTINE SUHDEM(Y,P,PR,RAT,K,PNOL,VAR,OMEAN,SD)
        IMPLICIT DOUBLE PRECISION (A-H,O-Z)
        DIMENSION Y(50),P(50),PR(50),W(50),DWE(50),DWA(50)
        RAT=0.
        CALL MAXB(ALFA,BETA,GAMMA,OMEAN,VAR,PNOL,Y,K,SD)
        WRITE (3,110) ALFA,BETA,GAMMA
110     FORMAT (' ALFA,BETA AND GAMMA' 3E20.10)
        CALL PROB(PR,ALFA,BETA,GAMMA,K)
        B=OMEAN/(VAR-OMEAN)
        CALL PARAM(B,ALFA,BETA,OMEAN,Y,K,SD)
        WRITE (3,120) ALFA,BETA
120     FORMAT (' ALFA AND BETA' 2E20.10)
        CALL PROBB(P,ALFA,BETA,K)
        DO 10 I=1,K
10      RAT=RAT+Y(I)*DLOG(P(I)/PR(I))
        RAT=-2.*RAT
        RETURN
        END
```

The main program for the likelihood ratio test of the hypothesis $H_o: \gamma = 0$ is LRATI

```
      IMPLICIT DOUBLE PRECISION (A-H,O-Z)
      DIMENSION X(50),TN(50),TP(50),Y(50),P(50),PR(50),W(50),DWE(50)
      DIMENSION DWA(50)
      CALL ASSIGN (2,'RE:',3)
      CALL ASSIGN (3,'WR:',3)
      WRITE (2,50)
50    FORMAT( ' GIVE THE NUMBER OF CLASSES')
      READ (2,10) K
10    FORMAT (I2)
      DO 35 I=1,K
      WRITE (2,52)
35    READ (2,30) Y(I)
52    FORMAT(' GIVE OBSERVATIONS IN THE CLASS')
30    FORMAT (F15.10)
      DO 1 I=1,K
1     X(I)=I-1.
      CALL STATM(K,X,Y,OMEAN,VAR,SKEW,PNOL,SD)
      CALL SUHDEM(Y,P,PR,RAT,K,PNOL,VAR,OMEAN,SD)
      DO 60 I=1,K
60    WRITE (3,55) PR(I),P(I)
55    FORMAT(' PO-NB PROB',E18.10,' NB PROB',E18.10)
      DO 71 I=1,K
      TP(I)=SD*PR(I)
71    TN(I)=SD*P(I)
      DO 72 I=1,K
72    WRITE (3,73) TP(I),TN(I),Y(I)
73    FORMAT (3F20.2)
      WRITE (3,56) RAT
56    FORMAT (' LOG-LIKELIHOOD IS',E20.10)
      STOP
      END
```

ON THE LIKELIHOOD RATIO TEST

In this section we consider some aspects on the validity of χ^2-test in testing the hypothesis $H_o: \gamma = 0$. The conditions for this asymptotic distribution of the likelihood ratio are given in sections 5f and 6e of [2]. There are two conditions that need a closer look. The others are quite straightforward.

First it is not clear from (1) that these are nonnegative, if we have negative gamma. In order that zero gamma were an inner point of the parameter space we need to establish that there exists such an $\varepsilon > 0$ that for $\gamma \in (-\varepsilon,\infty)$ makes (1) a probability function. That the sum of p_n's is equal to

unity is clear from the integral representation in [4]. If we now can show that $p_n \geq 0$ for all n, then they are also not larger than unity.

Now p_o is always nonnegative. The same is true for p_n when n is even. When n is uneven, then in (1) there is an even number of terms. We shall now prove that the sum of two subsequent terms is positive in an interval $(-\varepsilon,\infty)$. Taking the sum of the terms containing γ^{2k} and γ^{2k+1} we see that their sum is positive, if

$$\gamma > -(n-2k-1+\alpha)(2k+1)/((n-2k)(1+\beta)) = r(n,k),$$

where the equality is the definition of r(n,k). Now if $\alpha \geq 1$, then

$$r(n,k) \leq -(1+\beta)^{-1},$$

and if $\alpha < 1$, then

$$r(n,k) \leq -\alpha(1+\beta)^{-1},$$

these being valid for n=1,3,... and k=0,1,..., (n-1)/2. Hence, we can take $\varepsilon = (\min(1,\alpha))/(1+\beta)$.

The second task would be to prove that the information matrix J is positive definite when the restriction $\gamma = 0$ is valid. The elements of J are in general

$$J_{rs} = E\left(\frac{1}{p(x,\theta)} \frac{\partial p}{\partial \theta_r} \frac{}{p(x,\theta)} \frac{\partial p}{\partial \theta_s}\right),$$

if θ is the unknown parameter vector. We now take $\theta = (\alpha,\beta,\gamma)$. We have to compute the values of J_{rs} at points where $\gamma = 0$. We avoid largely the tedious calculations, and present here

only the main facts. Now let $p_n = p_n(\alpha,\beta,0)$ be the probability of n claims, if the parameter γ is zero, i.e., under the pure negative binomial distribution. In all the partial derivatives to follow are calculated at the points where $\gamma = 0$. The substitution of this is for short omitted. We have

$$\frac{\partial p_o}{\partial \gamma} = -p_o$$

$$\frac{\partial p_n}{\partial \beta} = p_n(n\beta+1-\alpha)/(n-1+\alpha), \quad n \geq 1$$

$$\frac{\partial p_n}{\partial \beta} = p_n(\alpha-n\beta)/(\beta(1+\beta))$$

$$\frac{\partial p_o}{\partial \alpha} = p_o \ln(\beta/(1+\beta)), \qquad n \geq 1$$

$$\frac{\partial p_n}{\partial \beta} = p_n[\sum_{i=1}^{n} (\alpha+i-1)^{-1} + \ln(\beta/(1+\beta))]$$

Substituting these into the definition of J_{rs} and using the above enumeration $\alpha \to 1, \beta \to 2$ and $\gamma \to 3$, we obtain

$$J_{11} = p_o(\ln\frac{1}{1+\beta})^2 + \sum_{n=1}^{\infty} (\sum_{i=1}^{n} (\alpha+i-1)^{-1} + \ln\frac{1}{1+\beta})^2 p_n$$

$$J_{12} = J_{21} = (\beta(1+\beta)^{-1} \sum_{i=1}^{\infty} (\alpha+i-1)^{-1} \sum_{n=i}^{\infty} (\alpha-\beta n)p_n$$

$$J_{13} = J_{31} = -p_o \ln\frac{\beta}{1+\beta} + \sum_{n=1}^{\infty} \frac{n\beta+1-\alpha}{n-1+\alpha}[\sum_{i=1}^{\infty} (\alpha+i-1)^{-1} + \ln\frac{\beta}{1+\beta}]p_n$$

$$J_{22} = \alpha/((1+\beta)\beta^2)$$

$$J_{23} = J_{32} = -1/(1+\beta)$$

$$J_{33} = p_o + \sum_{n=1}^{\infty} [(n\beta+1-\alpha)/(n-1+\alpha)^2]p_n.$$

Looking at the elements of J above it seems to us a hopeless task to obtain the positive definiteness of J in the general case. For this reason we have considered a special case, $\alpha = 1$. Then we have the simplified expressions

$$J_{11} = \frac{\beta}{1+\beta}(\ln\frac{\beta}{1+\beta})^2 + \sum_{n=1}^{\infty} [\sum_{i=1}^{n} \frac{1}{i} + \ln\frac{\beta}{1+\beta}]^2 \beta(1+\beta)^{-n-1}$$

$$J_{12} = J_{21} = -1/(\beta(1+\beta))$$

$$J_{13} = J_{31} = \beta\ln((1+\beta)/\beta)$$

$$J_{33} = \beta.$$

We now have $\beta > 0$ and

$$\begin{vmatrix} 1/((1+\beta)\beta^2) & -1/(1+\beta) \\ -1/(1+\beta) & \beta \end{vmatrix} > 0,$$

so that J is positive definite if we can show that the determinant of J is positive. This happens if

$$J_{11} > 1 - 2\beta\ln\frac{1+\beta}{\beta} + \beta(1+\beta)(\ln\frac{1+\beta}{\beta})^2. \qquad (10)$$

We have verified numerically that this is valid for reasonable values of β using various step lengths. The difference between the left and right hand sides seems to behave quite smoothly.

If we finally take $\alpha = c\beta$, $c > 0$, and let $\beta\uparrow\infty$, then all other $J_{ij} \to 0$ but J_{33}, which tends to $1/c$. The limiting J is not positive definite.

RESULTS OF THE COMPUTATIONS

In this section we present the complete lists of the computations concerning the fit of the convolution. Most of the results here were merely commented in [4]. We haved tried to fit our model on several real data found in ASTIN BULLETIN. We do not give here the references; these can be found in [4].

For each data the list contains sample mean, variance and third central moment. Additionally the zero class probability and number of observations are given. Then follow the maximum likelihood estimators for the parameters of the convolution and then for the negative binomial distribution. Then we give the class probabilities for both models beginning from the zero class and ending with the last class with positive observation. The following columns give the frequencies of these classes first for the convolution, then for the negative binomial distribution and then the observed frequencies. Then the test value for the likelihood ratio test is given (LOG-LIKELIHOOD). The two last values are the chi-square values with their degrees of freedom for the fits of the convolution and the negative binomial distribution in that order. Here the classes are joined together so that every theoretical frequency is at least five.
The results for those data (Muff and Delaporte) that lead to negative values for gamma are not presented here.

Tröbliger's data

```
MEAN     0.1442197634E 00

VARIANCE     0.1638699493E 00

3. CENTRAL MOMENT     0.2142856391E 00

ZERO CLASS PROBABILITY     0.8729492560E 00

NUMBER OF OBSERVATIONS  23589.

ALFA      0.2766327709E 00
BETA      0.3759793666E 01
GAMMA     0.7064318040E-01
ALFA FOR NB    0.1117895303E 01
BETA FOR NB    0.7751332249E 01
PO-NB PROB  0.8729436092E 00 NB PROB  0.8731509982E 00
PO-NB PROB  0.1124018149E 00 NB PROB  0.1115363207E 00
PO-NB PROB  0.1256599601E-01 NB PROB  0.1349635935E-01
PO-NB PROB  0.1743283873E-02 NB PROB  0.1602812549E-02
PO-NB PROB  0.2841810712E-03 NB PROB  0.1885488426E-03
PO-NB PROB  0.4990728885E-04 NB PROB  0.2205317336E-04
PO-NB PROB  0.9105722282E-05 NB PROB  0.2569494599E-05
             20591.87                20596.76                20592.00
              2651.45                 2631.03                 2651.00
               296.42                  318.37                  297.00
                41.12                   37.81                   41.00
                 6.70                    4.45                    7.00
                 1.18                    0.52                    0.00
                 0.21                    0.06                    1.00
LOG-LIKELIHOOD       0.3936361660E 01

CHI SQUARE IS         0.0042  DEGREES OF FREEDOM ARE    1

CHI SQUARE IS         3.5997  DEGREES OF FREEDOM ARE    2
```

Lemaire's data

```
MEAN       0.1010806364E 00

VARIANCE       0.1074478147E 00

3. CENTRAL MOMENT     0.1216465751E 00

ZERO CLASS PROBABILITY     0.9065567334E 00

NUMBER OF OBSERVATIONS 106974.
ALFA       0.5895314499E 00

BETA       0.9642598023E 01

GAMMA      0.3994239873E-01

ALFA FOR NB      0.1631274701E 01

BETA FOR NB      0.1613835012E 02

PO-NB PROB  0.9065456823E 00 NB PROB  0.9065830965E 00
PO-NB PROB  0.8642640667E-01 NB PROB  0.8629104080E-01
PO-NB PROB  0.6479008253E-02 NB PROB  0.6624191681E-02
PO-NB PROB  0.5036265585E-03 NB PROB  0.4678447166E-03
PO-NB PROB  0.4141571682E-04 NB PROB  0.3160627168E-04
              96976.82                      96980.82                      96978.00
               9245.38                       9230.90                       9240.00
                693.09                        708.62                        704.00
                 53.87                         50.05                         43.00
                  4.43                          3.38                          9.00
LOG-LIKELIHOOD ON     0.9634922789E 00
```

The fit of the three parameters of the convolution does not leave degrees of freedom for the chi-square test of fit. For the negative binomial distribution we have

```
CHI SQUARE IS          0.0908  DEGREES OF FREEDOM ARE     1
```

Thyrion's data

```
MEAN       0.2143536624E 00

VARIANCE      0.2889313713E 00

3. CENTRAL MOMENT    0.5406362987E 00

ZERO CLASS PROBABILITY    0.8286650460E 00

NUMBER OF OBSERVATIONS    9461.
ALFA       0.2006136757E 00

BETA       0.1666513484E 01

GAMMA      0.9397439298E-01

ALFA FOR NB     0.7015121904E 00

BETA FOR NB     0.3272685815E 01

PO-NB PROB   0.8283906275E 00 NB PROB   0.8294061324E 00
PO-NB PROB   0.1401710197E 00 NB PROB   0.1361762923E 00
PO-NB PROB   0.2354541809E-01 NB PROB   0.2711475071E-01
PO-NB PROB   0.5568068260E-02 NB PROB   0.5714659174E-02
PO-NB PROB   0.1594201369E-02 NB PROB   0.1237680555E-02
PO-NB PROB   0.4929920371E-03 NB PROB   0.2723799721E-03
PO-NB PROB   0.1586080444E-03 NB PROB   0.6057773332E-04
PO-NB PROB   0.5233603588E-04 NB PROB   0.1357334325E-04
             7837.40                 7847.01                  7840.00
             1326.16                 1288.36                  1317.00
              222.76                  256.53                   239.00
               52.68                   54.07                    42.00
               15.08                   11.71                    14.00
                4.66                    2.58                     4.00
                1.50                    0.57                     4.00
                0.50                    0.13                     1.00
LOG-LIKELIHOOD          0.9529057177E 01


CHI SQUARE IS         4.1205  DEGREES OF FREEDOM ARE    2


CHI SQUARE IS         8.7661  DEGREES OF FREEDOM ARE    2
```

Pesonen's data

```
MEAN      0.8766824300E-01

VARIANCE      0.9818881333E-01

3. CENTRAL MOMENT    0.1288312947E 00

ZERO CLASS PROBABILITY    0.9199708985E 00

NUMBER OF OBSERVATIONS    5498.
ALFA      0.1135839346E 00

BETA      0.3397984276E 01

GAMMA     0.5424138596E-01

ALFA FOR NB     0.8195103463E 00

BETA FOR NB     0.9347858681E 01

PO-NB PROB  0.9198525044E 00 NB PROB  0.9200854541E 00
PO-NB PROB  0.7365051861E-01 NB PROB  0.7286720590E-01
PO-NB PROB  0.5649348786E-02 NB PROB  0.6406283615E-02
PO-NB PROB  0.7043480078E-03 NB PROB  0.5818460770E-03
PO-NB PROB  0.1167945454E-03 NB PROB  0.5369147327E-04
PO-NB PROB  0.2137802163E-04 NB PROB  0.5001355718E-05
            5057.35               5058.63                  5058.00
             404.93                400.62                   403.00
              31.06                 35.22                    34.00
               3.87                  3.20                     2.00
               0.64                  0.30                     0.00
               0.12                  0.03                     1.00
LOG-LIKELIHOOD       0.1175953085E 01
```

No degrees of freedom for either model is left.

Bühlmann's data

MEAN 0.1551400466E 00

VARIANCE 0.1793155390E 00

3. CENTRAL MOMENT 0.2394591053E 00

ZERO CLASS PROBABILITY 0.8652599434E 00

NUMBER OF OBSERVATIONS 119853.
ALFA 0.4001495974E 00

BETA 0.4068437434E 01

GAMMA 0.5678543159E-01

ALFA FOR NB 0.1032668356E 01

BETA FOR NB 0.6656362294E 01

PO-NB PROB	0.8652578390E 00	NB PROB	0.8654235575E 00		
PO-NB PROB	0.1174455422E 00	NB PROB	0.1167258664E 00		
PO-NB PROB	0.1470963117E-01	NB PROB	0.1549462826E-01		
PO-NB PROB	0.2161727949E-02	NB PROB	0.2045796119E-02		
PO-NB PROB	0.3520355160E-03	NB PROB	0.2693843681E-03		
PO-NB PROB	0.6027793932E-04	NB PROB	0.3541426418E-04		
PO-NB PROB	0.1061695392E-04	NB PROB	0.4650653115E-05		
	103703.75		103723.61		103704.00
	14076.20		13989.95		14075.00
	1762.99		1857.08		1766.00
	259.09		245.19		255.00
	42.19		32.29		45.00
	7.22		4.24		6.00
	1.27		0.56		2.00

LOG-LIKELIHOOD 0.1155354665E 02

CHI SQUARE IS 0.3252 DEGREES OF FREEDOM ARE 2

CHI SQUARE IS 12.1187 DEGREES OF FREEDOM ARE 2

Tröbliger's and Bühlmann's data joined together

```
MEAN       0.1533442088E 00

VARIANCE       0.1767907417E 00

3. CENTRAL MOMENT      0.2353892777E 00

ZERO CLASS PROBABILITY     0.8665244489E 00

NUMBER OF OBSERVATIONS 143442.
ALFA      0.3783791287E 00

BETA      0.4018212045E 01

GAMMA     0.5917816568E-01

ALFA FOR NB      0.1043090034E 01

BETA FOR NB      0.6802278628E 01

PO-NB PROB   0.8665219127E 00 NB PROB   0.8666948501E 00
PO-NB PROB   0.1166159553E 00 NB PROB   0.1158688127E 00
PO-NB PROB   0.1435701554E-01 NB PROB   0.1517059489E-01
PO-NB PROB   0.2092969241E-02 NB PROB   0.1972307848E-02
PO-NB PROB   0.3408967082E-03 NB PROB   0.2555092745E-03
PO-NB PROB   0.5858471309E-04 NB PROB   0.3303025532E-04
PO-NB PROB   0.1037271750E-04 NB PROB   0.4263814378E-05
          124295.64                 124320.44                 124296.00
           16727.63                  16620.45                  16726.00
            2059.40                   2176.10                   2063.00
             300.22                    282.91                    296.00
              48.90                     36.65                     52.00
               8.40                      4.74                      6.00
               1.49                      0.61                      3.00
LOG-LIKELIHOOD        0.1532609531E 02


  CHI SQUARE IS           0.4081  DEGREES OF FREEDOM ARE      2


  CHI SQUARE IS          15.9166  DEGREES OF FREEDOM ARE      3
```

Lemaire's and Bühlmann's data joined together

MEAN 0.1296450599E 00

VARIANCE 0.1461495936E 00

3. CENTRAL MOMENT 0.1867999334E 00

ZERO CLASS PROBABILITY 0.8847359441E 00

NUMBER OF OBSERVATIONS 226827.
ALFA 0.3196597022E 00

BETA 0.4405432512E 01

GAMMA 0.5708471506E-01

ALFA FOR NB 0.1061570053E 01

BETA FOR NB 0.8188280012E 01

```
PO-NB PROB  0.8847269342E 00 NB PROB  0.8848657875E 00
PO-NB PROB  0.1028242590E 00 NB PROB  0.1022331731E 00
PO-NB PROB  0.1081475701E-01 NB PROB  0.1146900441E-01
PO-NB PROB  0.1390819686E-02 NB PROB  0.1273838716E-02
PO-NB PROB  0.2048334055E-03 NB PROB  0.1407713188E-03
PO-NB PROB  0.3213881617E-04 NB PROB  0.1550940744E-04
PO-NB PROB  0.5216718084E-05 NB PROB  0.1705276714E-05
```

200679.96	200711.45	200682.00
23323.32	23189.24	23315.00
2453.08	2601.48	2470.00
315.48	288.94	298.00
46.46	31.93	54.00
7.29	3.52	6.00
1.18	0.39	2.00

LOG-LIKELIHOOD 0.2176679945E 02

CHI SQUARE IS 2.3686 DEGREES OF FREEDOM ARE 2

CHI SQUARE IS 26.6239 DEGREES OF FREEDOM ARE 2

Lemaire's and Bühlmann's data joinec

MEAN 0.1296450599E 00

VARIANCE 0.1461495936E 00

3. CENTRAL MOMENT 0.1867999334E

ZERO CLASS PROBABILITY 0.8847359

NUMBER OF OBSERVATIONS 226827.
ALFA 0.3196597022E 00

BETA 0.4405432512E 01

GAMMA 0.5708471506E-01

ALFA FOR NB 0.1061570053E 01

BETA FOR NB 0.8188280012E 01

```
PO-NB PROB  0.8847269342E 00 NB PRO
PO-NB PROB  0.1028242590E 00 NB PRO
PO-NB PROB  0.1081475701E-01 NB PRO
PO-NB PROB  0.1390819686E-02 NB PRO
PO-NB PROB  0.2048334055E-03 NB PRC
PO-NB PROB  0.3213881617E-04 NB PRC
PO-NB PROB  0.5216718084E-05 NB PRC
          200679.96           20071
           23323.32            2318
            2453.08             260
             315.48              28
              46.46               3
               7.29
               1.18
LOG-LIKELIHOOD        0.2176679945E
```

CHI SQUARE IS 2.3686 DEGREE

CHI SQUARE IS 26.6239 DEGRE